

Sorting and Searching by Distribution: From Generic Discrimination to Generic Tries

Fritz Henglein¹ and Ralf Hinze²

¹ Department of Computer Science, University of Copenhagen
henglein@diku.dk

² Department of Computer Science, University of Oxford
ralf.hinze@cs.ox.ac.uk

Abstract. A discriminator partitions values associated with keys into groups listed in ascending order. Discriminators can be defined generically by structural recursion on representations of ordering relations. Employing type-indexed families we demonstrate how tries with an optimal-time lookup function can be constructed generically in worst-case linear time. We provide generic implementations of comparison, sorting, discrimination and trie building functions and give equational proofs of correctness that highlight core relations between these algorithms.

1 Introduction

Sorting and searching are some of the most fundamental topics in computer science. In this paper we define *generic* functions for solving sorting and searching problems, based on *distributive*, that is “radix-sort-like”, techniques rather than comparison-based techniques. The functions are indexed by representations of ordering relations on keys of type K . In each case the input is an association list of key-value pairs, and the values are treated as *satellite data*, that is, the functions are *parametric* in the value type V . Intuitively, this means values are pointers that are not dereferenced during execution of these functions [1]. We identify a hierarchy of operations:³

$$\begin{aligned} \text{sort} &:: \text{Order } k \rightarrow [k \times v] \rightarrow [v] \\ \text{discr} &:: \text{Order } k \rightarrow [k \times v] \rightarrow [[v]] \\ \text{trie} &:: \text{Order } k \rightarrow [k \times v] \rightarrow \text{Trie } k [v] \end{aligned}$$

The *sorting function*, *sort*, outputs the value components according to the given order on K without, however, returning the key component. For example,

$$\begin{aligned} &\gg \text{sort } (\text{OList } \text{OChar}) [(\text{"ab"}, 1), (\text{"ba"}, 2), (\text{"abc"}, 3), (\text{"ba"}, 4)] \\ &[1, 3, 2, 4] \end{aligned}$$

³ Executable code is rendered in Haskell, which requires lower-case identifiers for type variables. We use the corresponding upper-case identifiers in the running text and in program calculations.

where *OList OChar* denotes the standard lexicographic order on strings. We require that *sort* be stable in the sense that the relative order of values with equivalent keys is preserved. Discarding the keys may seem surprising and restrictive at first. Nothing is lost, however, since parametricity allows us to arrange it so that the keys are also returned. We simply associate the keys with themselves.

```
>> sort (OList OChar) [("ab", "ab"), ("ba", "ba"), ("abc", "abc"), ("ba", "ba")]
["ab", "abc", "ba", "ba"]
```

The *discriminator*, *discr*, outputs the value components grouped into runs of values associated with equivalent keys. For example,

```
>> discr (OList OChar) [("ab", 1), ("ba", 2), ("abc", 3), ("ba", 4)]
[[1], [3], [2, 4]] .
```

The *trie constructor*, *trie*, outputs a trie that can subsequently be efficiently searched for values associated to a particular key. The type of trie constructed depends on the type of the keys. For example,

```
>> let t = trie (OList OChar) [("ab", 1), ("ba", 2), ("abc", 3), ("ba", 4)]
>> lookup t "ba"
Just [2, 4] .
```

The function *discr* was introduced by Henglein [2, 3] (originally called *discr*). It provides a framework for bootstrapping any base sorting algorithm for a *finite* type, such as bucket sort, to a large class of user-definable orders on first-order and recursive types. To this end it employs a strategy corresponding to most-significant-digit (MSD) in radix sorting.

The functions *sort* and *trie* are novel. Algorithmically, *sort* does the same as *discr*, but employing a least-significant-digit (LSD) strategy. Drawing on the informal correspondence of MSD radix sort with tries [4, p. 3], *trie* generalizes *discr* and generates the generalized tries introduced by Hinze [5]. It subsumes *discr* (which in turn subsumes *sort*) in the sense that it executes in the same time (usually linear in the size of the input keys), but additionally facilitates efficient search for values associated with any key.


In this paper we make the following novel contributions:

- We show that a function of type $[K \times V] \rightarrow [V]$ is a *stable sorting function* if and only if it is *strongly natural* in V , preserves singleton lists, and sorts lists of length 2 correctly. A function is strongly natural if it commutes with filtering, that is, the removal of elements from a list.
- We give new *generic* definitions of: *sort*, which generalizes least-significant-digit (LSD) radix sort to arbitrary types and orders definable by an expressive language of *order representations*; and *trie*, which generalizes *discr* to construct efficiently key-searchable tries. Both run in worst-case linear time for a large class of orders.
- We provide equational proofs for *sort* o being a stable sorting function and show that $\text{sort } o = \text{concat} \cdot \text{discr } o$ and $\text{discr } o = \text{flatten} \cdot \text{trie } o$ for all

inductively defined order representations o , where *concat* is list concatenation and *flatten* lists the values stored in a trie in ascending key order. The first equality is nontrivial as *discr* and *sort* have different underlying algorithmic strategies for product types: MSD versus LSD. The proof highlights the strong naturality properties of *sort* and *discr*.

- We offer preliminary benchmark results of our generic distributive sorting functions, which are surprisingly promising when compared to Haskell’s built-in comparison-based sorting function.

The paper focuses on and highlights the core relations between these algorithms, notably the role of *strong naturality*. Here we limit ourselves to a restricted class of orders and leave asymptotic analysis, performance engineering, and a proper empirical performance analysis for future work. But certainly some benchmarks are not amiss to whet the appetite. The task is to sort the words of Project Gutenberg’s The Bible, King James Version (5218802 characters, 824337 words). We compare Haskell’s built-in *sortBy* called with Haskell’s own *compare* and our generically defined comparison function *cmp o*, to generic sorting and generic discrimination, and to sorting via generic tries.



We assume familiarity with the programming language GHC Haskell and basic notions of category theory. Unless noted otherwise, we work in **Set**, the category of sets and total functions.

2 Order Representations

Comparison-based sorting and searching methods are attractive because they easily generalize to arbitrary orders: simply parameterize the program code for, say, Quicksort [6] over its comparison function, and apply it to a user-defined ordering $leq :: T \rightarrow T \rightarrow \mathbb{B}$. An analogous approach works for searching on T using, say, red-black trees [7, 8]. While maximally expressive, specifying orders via such “black-box” binary comparisons, has two disadvantages:

1. Deliberately or erroneously, *leq* may not implement a total preorder.
2. Both sorting and searching are subject to lower bounds on their performance: sorting requires $\Omega(n \log n)$ comparisons, and searching for a key requires $\Omega(\log n)$, where n is the number of keys in the input.

However, theoretically and practically faster *distributive* methods are known for *certain* orders, notably radix sorting and tries.

As shown by Henglein [3], many orders can be denoted by *order representations*, constructors for building new orders from old:

```

data Order :: * → * where
  OUnit :: Order ()
  OSum  :: Order k1 → Order k2 → Order (k1 + k2)
  OProd :: Order k1 → Order k2 → Order (k1 × k2)
  OMap  :: (k1 → k2) → (Order k2 → Order k1)
  OChar :: Order Char -- 7 bit ASCII .

```

Here *OUnit* denotes the trivial order on the unit type. *OSum* o_1 o_2 represents the lexicographic order on tagged values such that *Inl*-tagged values are less than *Inr*-tagged values, and values with the same tag are ordered by o_1 or o_2 , depending on the tag. *OProd* o_1 o_2 denotes the lexicographic order on pairs, ordering pairs according to their first component, where pairs with equivalent first component are ordered according to their second component. *OMap* f o orders the domain of f according to the order o on its codomain. Note that *OMap* is contravariant. Finally, *OChar* denotes the standard order on 7-bit ASCII characters.

The *OMap*-constructor adds considerable expressiveness. For example,

```

rprod :: Order k1 → Order k2 → Order (k1 × k2)
rprod o1 o2 = OMap (λ(a, b) → (b, a)) (OProd o2 o1)

```

specifies the lexicographic order on pairs based on the *second* component as the dominant one. Similarly, *rsum* can be specified, which orders *Inr*-tagged values as less than *Inl*-tagged ones.

Order representations are terms that can be treated *inductively* as finite trees or *coinductively* as potentially infinite trees.

The coinductive approach permits definition of orders for recursive data types by guarded recursion. For example,

```

olist :: Order k → Order [k]
olist o = os where os = OMap out (OSum OUnit (OProd o os))
out :: [a] → () + a × [a]
out []      = Inl ()
out (a : as) = Inr (a, as)

```

defines the standard lexicographic order on lists, based on the element order o . Henglein [3] takes the coinductive approach with additional order constructors for inverse, multiset and set orders.

In the inductive approach we can add new constructors explicitly:

```

OList :: Order k → Order [k]

```

or, more generally, employ an explicit fixed point operator [2].

The expressiveness of order representations is orthogonal to the aims of this paper. For simplicity we assume the inductive approach and concentrate on sums and products.

3 Generic Comparison

We require order representations to denote *total preorders*.

```

leq :: Order k → (k → k → ℬ)
leq OUnit a b      = True
leq (OSum o1 o2) a b = case (a, b) of
    (Inl a1, Inl a2) → leq o1 a1 a2
    (Inl –, Inr –)   → True
    (Inr –, Inl –)   → False
    (Inr b1, Inr b2) → leq o2 b1 b2
leq (OProd o1 o2) a b = leq o1 (fst a) (fst b) ∧
    (leq o1 (fst b) (fst a) ⇒ leq o2 (snd a) (snd b))
leq (OMap g o) a b   = leq o (g a) (g b)
leq (OChar) a b     = a ≤ b
    
```

$leq\ o$ is indeed a total preorder; it is transitive and total, $leq\ o\ x\ y \vee leq\ o\ y\ x$. Because of totality the case for $OProd$ can also be written as

```

leq (OProd o1 o2) a b = if leq o2 (snd a) (snd b) then leq o1 (fst a) (fst b)
    else ¬ (leq o1 (fst b) (fst a))
    
```

This variant is strict ($leq\ o_1$ and $leq\ o_2$ are called), but it calls $leq\ o_1$ only once. The first variant is lazy ($leq\ o_2$ is not necessarily called), but possibly calls $leq\ o_1$ twice.

The function leq implements a two-way comparison; a more useful function is $cmp :: Order\ k \rightarrow (k \rightarrow k \rightarrow Ordering)$, which implements a three-way comparison and avoids the double traversal in the product case.

4 Generic Distributive Sorting

Generic sorting takes a list of key-value pairs and returns the values in non-decreasing order of their associated keys. The keys are discarded in the course of this process. The idea is that, barring $OMap$ in order representations, each component of each key is touched *exactly* once. Consequently, the running time of $sort$ is proportional to the total size of the keys (again, ignoring $OMap$).

```

sort :: Order k → [k × v] → [v]
sort o      [] = []
sort (OUnit) rel = map val rel
sort (OSum o1 o2) rel = sort o1 (filter froml rel) ++ sort o2 (filter fromr rel)
sort (OProd o1 o2) rel = sort o1 (sort o2 (map curryr rel))
sort (OMap g o) rel = sort o (map (g × id) rel)
sort (OChar) rel = bucketSort (' \NUL ', ' \DEL ') rel
    
```

Like generic comparison, $sort$ is indexed by order representations. It is furthermore parametric in the type of values. Let us discuss each case in turn.

The first equation is vital for the coinductive approach to recursive types. It is necessary to ensure that for each recursive invocation of *sort* the total size of the keys is strictly decreasing.

For the unit type there is little to do: we simply discard the keys using *val* defined $val (k, v) = v$.

The case for sums takes an approach à la Quicksort: the input list is partitioned into a list whose keys are of the form *Inl* k_1 and a second list whose keys are of the form *Inr* k_2 . The constructors are discarded, the sub-lists are sorted recursively using the appropriate orders, and the final results are concatenated. (As an aside, in the partitioning phase we touch the keys actually twice, but this is easily avoided by combining the two sweeps into a single one.) The function $filter :: (a \rightarrow \text{Maybe } b) \rightarrow ([a] \rightarrow [b])$, called *mapMaybe* elsewhere, combines mapping and filtering: if the argument function returns a value of the form *Just* b , then b is included in the output list. If the result is *Nothing*, well, nothing is added. (Don't confuse our *filter* with Haskell's $filter :: (a \rightarrow \mathbb{B}) \rightarrow ([a] \rightarrow [a])$).

$$\begin{aligned} filter &:: (a \rightarrow \text{Maybe } b) \rightarrow ([a] \rightarrow [b]) \\ filter\ p\ xs &= [y \mid x \leftarrow xs, \text{Just } y \leftarrow [p\ x]] \end{aligned}$$

The function $froml :: (k_1 + k_2 \times v) \rightarrow \text{Maybe } (k_1 \times v)$ maps $(\text{Inl } k_1, v)$ to *Just* (k_1, v) , and $(\text{Inr } k_2, v)$ to *Nothing*. The function *fromr* is defined analogously.

The most interesting case is the one for products. The natural isomorphism $curryr : (K_1 \times K_2) \times V \cong K_2 \times (K_1 \times V)$ shifts the *more significant* part of the key into the value component. Then *sort* is called twice: the first invocation sorts according to o_2 discarding the K_2 part, the second sorts according to o_1 discarding the K_1 component. For this to be correct, *sort* o_1 had better be stable; we shall return to this point below. Furthermore, *sort* relies on *polymorphic recursion*: the first call to *sort* instantiates V to $K_1 \times V$.

For *OMap* we simply apply the key transformation using $map (g \times id)$ and then sort the transformed keys-value pairs.

Characters are sorted using bucket sort, which can be seen as a specialization of *sort* (actually of *discr* introduced in Section 5) for enumeration types.

$$\begin{aligned} bucketSort &:: (\text{Bounded } i, \text{Ix } i) \Rightarrow (i, i) \rightarrow [(i, v)] \rightarrow [v] \\ bucketSort\ bs\ rel &= concat (elems (accumArray (\lambda ws\ w \rightarrow ws \# [w]) [] bs\ rel)) \end{aligned}$$

(Here $\#$ is used for clarity; in our implementation it is replaced by a constant-time operation.) Any other algorithm for sorting characters or, for that matter, other primitive types could be used. The particular algorithm invoked can even be made *data dependent*; for small *rel* we might choose insertion sort instead of bucket sort to avoid sparse bucket table traversals. The key point is that *sort* reduces a sorting problem to basic sorting on finite domains. Conversely, it extends distributive sorting from their restricted domains such as small integers and character strings to arbitrary orders definable by order representations.

Let us now turn to the correctness of *sort*. Its implementation builds upon standard components, except perhaps the case for sums, which relies on the function *filter*. Note that *filter* takes a partial function as an argument, represented

by a total function of type $A \rightarrow \text{Maybe } B$, an arrow in the Kleisli category induced by the monad `Maybe`. For the proofs it will be convenient to actually work in this Kleisli category (but only for the arguments of `filter`). In the calculations, we signal these steps by the hint “Kleisli:”. A few remarks are in order.

Working in the Kleisli category has the advantage that the notation is fairly light-weight: we write `id` rather than η , and we write $p \cdot q$ for the Kleisli composition of p and q rather than $\mu \cdot \text{Maybe } p \cdot q$. We also silently embed total functions into the Kleisli category: `filter f` really means `filter (\eta \cdot f)`. In any case, there is little room for confusion since we let $f, g \dots$ range over total functions and p, q over partial functions. The product \times can be lifted to a binary functor \otimes over the Kleisli category, which is, however, not a categorical product. Rather, \otimes is a so-called tensor product, a binary functor which is coherently associative and commutative. We overload \times to denote both the product in the underlying category and the tensor product.

A partial function that we will use time and again is `inlo`, the left-inverse of `inl`. The partial function `froml` can be neatly expressed in terms of `inlo`: we have `froml = inlo \otimes id`, or just `inlo \times id`. Likewise, `fromr = inro \otimes id`.

The function `filter` satisfies a variety of properties. First and foremost, it is a monoid homomorphism:

$$\text{filter } p [] = [] \quad , \quad (4.1a)$$

$$\text{filter } p (xs \# ys) = \text{filter } p xs \# \text{filter } p ys \quad . \quad (4.1b)$$

Furthermore, `filter` is functorial, taking Kleisli arrows to arrows in the underlying category. Formally, `filter` is the arrow part of the functor `Filter : Kleisli \rightarrow Set`, whose object part is defined `Filter A = List A`. In other words, `filter` preserves identity and composition.

$$\text{filter } id = id \quad (4.2a)$$

$$\text{filter } (p \cdot q) = \text{filter } p \cdot \text{filter } q \quad (4.2b)$$

Moreover, if its first argument is a total function, then `filter f` is just `List f`.

Most of the following proofs will be conducted in a point-free style. For reference, here is a suitably reworked version of `sort`.

$$\begin{aligned} \text{sort } (OUnit) &= \text{List } val \\ \text{sort } (OSum o_1 o_2) &= \text{sort } o_1 \cdot \text{filter } (inl^o \times id) \# \text{sort } o_2 \cdot \text{filter } (inr^o \times id) \\ \text{sort } (OProd o_1 o_2) &= \text{sort } o_1 \cdot \text{sort } o_2 \cdot \text{List } currry \end{aligned}$$

The sum case uses a lifted variant of `append`, $(f \# g) x = f x \# g x$, overloading the operator $\#$ to denote both the lifted and the unlifted version.

4.1 Naturality

A vital property of `sort o` is that it is natural in the type of values, `sort o : List o (K \times) \rightarrow List`, that is

$$\text{List } f \cdot \text{sort } o = \text{sort } o \cdot \text{List } (id \times f) \quad , \quad (4.3)$$

for all $f: A \rightarrow B$.

Because of naturality it is sufficient to show that the instance $sort\ o : [K \times \mathbb{N}] \rightarrow [\mathbb{N}]$ works correctly. (Recall that values can be seen as pointers; natural numbers are like unique pointers.) Formally, $sort\ o$ is fully determined by this instance:

$$\begin{aligned}
& sort\ o\ [(k_1, v_1), \dots, (k_n, v_n)] \\
= & \{ \text{let } ix : \mathbb{N} \rightarrow V \text{ be an indexing function so that } ix\ i = k_i \} \\
& sort\ o\ (List\ (id \times ix)\ [(k_1, 1), \dots, (k_n, n)]) \\
= & \{ \text{sort is natural (4.3)} \} \\
& List\ ix\ (sort\ o\ [(k_1, 1), \dots, (k_n, n)]) .
\end{aligned}$$

In the last equation $sort\ o$ is used at instance \mathbb{N} .

Of course, the statement that $sort\ o$ is natural requires proof. Actually, $sort$ satisfies a much stronger property, which we discuss next.

4.2 Strong Naturality

Property (4.3) remains valid if we replace `List` by `filter`:

$$filter\ p \cdot sort\ o = sort\ o \cdot filter\ (id \times p) , \quad (4.4)$$

for all $p : A \rightarrow \text{Maybe } B$. It does not matter whether we filter before or after an invocation of $sort\ o$, as long as $filter$ only refers to the values, and not the keys.

Now, since $filter$ is the arrow part of a functor between the Kleisli category of `Maybe` and the underlying category, (4.4) also amounts to a naturality property, $sort\ o : \text{Filter} \circ (K \otimes) \rightarrow \text{Filter}$. A simple consequence of what we call strong naturality (4.4) is that $sort$ preserves the empty list.

Turning to the proof of (4.4), we proceed by induction over the structure of order representations.

Case $o = OUnit$:

$$\begin{aligned}
& filter\ p \cdot sort\ OUnit \\
= & \{ \text{definition of } sort \} \\
& filter\ p \cdot List\ val \\
= & \{ \text{Kleisli: } p \cdot val = val \cdot (id \times p) \} \\
& List\ val \cdot filter\ (id \times p) \\
= & \{ \text{definition of } sort \} \\
& sort\ OUnit \cdot filter\ (id \times p)
\end{aligned}$$

Recall that \times aka \otimes is *not* a categorical product in the Kleisli category, we have $val \cdot (id \times p) = p \cdot val$, but *not* $key \cdot (id \times p) = id \cdot key$ where $key\ (k, v) = k$.

Case $o = OSum\ o_1\ o_2$: the central step is the fourth one, where we swap two filters, one that acts on the keys and a second that acts on the values.

$$\begin{aligned}
& filter\ p \cdot sort\ (OSum\ o_1\ o_2) \\
= & \{ \text{definition of } sort \} \\
& filter\ p \cdot (sort\ o_1 \cdot filter\ (inl^p \times id) \# \dots) \\
= & \{ \text{filter is a monoid homomorphism (4.1b)} \} \\
& filter\ p \cdot sort\ o_1 \cdot filter\ (inl^p \times id) \# \dots \\
= & \{ \text{ex hypothesi} \}
\end{aligned}$$

$$\begin{aligned}
 & \text{sort } o_1 \cdot \text{filter } ((id + id) \times p) \cdot \text{filter } (inl^\circ \times id) \# \dots \\
 = & \{ \text{Kleisli: } \times \text{ is a binary functor } \} \\
 & \text{sort } o_1 \cdot \text{filter } (inl^\circ \times id) \cdot \text{filter } ((id + id) \times p) \# \dots \\
 = & \{ \text{fusion: } f \cdot h \# g \cdot h = (f \# g) \cdot h \} \\
 & (\text{sort } o_1 \cdot \text{filter } (inl^\circ \times id) \# \dots) \cdot \text{filter } ((id + id) \times p) \\
 = & \{ \text{definition of } \text{sort} \} \\
 & \text{sort } (OSum \ o_1 \ o_2) \cdot \text{filter } ((id + id) \times p)
 \end{aligned}$$

Note that the lifting of $+$ is a categorical coproduct in the Kleisli category.

Case $o = OProd \ o_1 \ o_2$: note that Property (4.4) is universally quantified over all p . This is essential as sort relies on polymorphic recursion: the second use of the induction hypothesis $(*)$ instantiates p to $id \times p$.

$$\begin{aligned}
 & \text{filter } p \cdot \text{sort } (OProd \ o_1 \ o_2) \\
 = & \{ \text{definition of } \text{sort} \} \\
 & \text{filter } p \cdot \text{sort } o_1 \cdot \text{sort } o_2 \cdot \text{List } \text{curryr} \\
 = & \{ \text{ex hypothesi} \} \\
 & \text{sort } o_1 \cdot \text{filter } (id \times p) \cdot \text{sort } o_2 \cdot \text{List } \text{curryr} \\
 = & \{ \text{ex hypothesi } (*) \} \\
 & \text{sort } o_1 \cdot \text{sort } o_2 \cdot \text{filter } (id \times (id \times p)) \cdot \text{List } \text{curryr} \\
 = & \{ \text{Kleisli: } (q \times (p \times r)) \cdot \text{curryr} = \text{curryr} \cdot ((p \times q) \times r) \} \\
 & \text{sort } o_1 \cdot \text{sort } o_2 \cdot \text{List } \text{curryr} \cdot \text{filter } ((id \times id) \times p) \\
 = & \{ \text{definition of } \text{sort} \} \\
 & \text{sort } (OProd \ o_1 \ o_2) \cdot \text{filter } ((id \times id) \times p)
 \end{aligned}$$

The natural isomorphisms $\text{val}: 1 \times K \cong K$ and $\text{curryr}: (K_1 \times K_2) \times V \cong K_2 \times (K_1 \times V)$ are also natural isomorphisms in the Kleisli category.

4.3 Correctness: Permutation

Our first goal is to show that $\text{sort } o$ produces a permutation of the input values. Perhaps surprisingly, it suffices to show that $\text{sort } o$ permutes one-element lists! We already know that it is sufficient to show the correctness of a particular instance: $\text{sort } o [(k_1, 1), \dots, (k_n, n)]$. Now, let λi be the partial function that maps i to i and is undefined otherwise. Let $1 \leq i \leq n$, then

$$\begin{aligned}
 & \text{filter } \lambda i \text{ } (\text{sort } o [(k_1, 1), \dots, (k_n, n)]) \\
 = & \{ \text{sort is strongly natural (4.4)} \} \\
 & \text{sort } o (\text{filter } (id \times \lambda i) [(k_1, 1), \dots, (k_n, n)]) \\
 = & \{ \text{filter } (id \times \lambda i) [(k_1, 1), \dots, (k_n, n)] = [(k_i, i)] \} \\
 & \text{sort } o [(k_i, i)] \\
 = & \{ \text{proof obligation: } \text{sort} \text{ permutes 1-element lists (4.5)} \} \\
 & [i] .
 \end{aligned}$$

Thus, $\text{sort } o$ outputs each index exactly once; in other words, it permutes the input list. The proof obligation (recall that $\text{return } a = [a]$)

$$\text{sort } o \cdot \text{return} = \text{return} \cdot \text{val} \tag{4.5}$$

is easy to discharge and left as an instructive exercise to the reader.

4.4 Correctness: Ordered

Our second task is to show that the values are output in non-decreasing order of their associated keys. We aim to show

$$\text{sort } o [\dots, (k_i, i), \dots, (k_j, j), \dots] = [\dots, i, \dots, j, \dots] \iff \text{leq } o k_i k_j .$$

Due to strong naturality, it suffices to show that *sort o* works correctly on two-element lists! Let $\lambda i, j \rfloor$ be the partial function that maps i to i and j to j and is undefined otherwise. Let $1 \leq i, j \leq n$, then

$$\begin{aligned} & \text{filter } \lambda i, j \rfloor (\text{sort } o [(k_1, 1), \dots, (k_n, n)]) = [i, j] \\ \iff & \{ \text{sort is strongly natural (4.4)} \} \\ & \text{sort } o (\text{filter } (id \times \lambda i, j \rfloor) [(k_1, 1), \dots, (k_n, n)]) = [i, j] \\ \iff & \{ \text{filter } (id \times \lambda i, j \rfloor) [(k_1, 1), \dots, (k_n, n)] = [(k_i, i), (k_j, j)] \} \\ & \text{sort } o [(k_i, i), (k_j, j)] = [i, j] \\ \iff & \{ \text{proof obligation: sort sorts 2-element lists (4.6)} \} \\ & \text{leq } o k_i k_j . \end{aligned}$$

Thus, *sort o* outputs i before j if and only if $\text{leq } o k_i k_j$. Since we already know that *sort o* permutes its input, this implies the correctness of *sort o*.

It remains to show that *sort* treats 2-element lists correctly: let $i \neq j$, then

$$\text{sort } o [(a, i), (b, j)] = [i, j] \iff \text{leq } o a b . \quad (4.6)$$

Since only a small finite number of cases have to be considered, this is a simple exercise, which we relegate to Appendix A.

5 Generic Discrimination

A discriminator returns a list of non-empty lists of values, where the inner lists group values whose keys are equivalent. Again, the keys are discarded in the process, but, barring *OMap* in order representations, this time each component of each key is touched *at most* once.

$$\begin{aligned} \text{discr} &:: \text{Order } k \rightarrow [k \times v] \rightarrow [[v]] \\ \text{discr } o & \quad [] = [] \\ \text{discr } o & \quad [(k, v)] = [[v]] \\ \text{discr } OUnit & \quad \text{rel} = [\text{map val rel}] \\ \text{discr } (OSum o_1 o_2) & \quad \text{rel} = \text{discr } o_1 (\text{filter froml rel}) \text{ ++ } \text{discr } o_2 (\text{filter fromr rel}) \\ \text{discr } (OProd o_1 o_2) & \quad \text{rel} = \text{concat } (\text{map } (\text{discr } o_2) (\text{discr } o_1 (\text{map curryl rel}))) \\ \text{discr } (OMap g o) & \quad \text{rel} = \text{discr } o (\text{map } (g \times id) \text{ rel}) \\ \text{discr } (OChar) & \quad \text{rel} = \text{bucketDiscr } (\text{'\NUL'}, \text{'\DEL'}) \text{ rel} \end{aligned}$$

In the unit case we have identified a group of key-value pairs whose keys are equivalent, even identical. This group is returned as a singleton list. The sum case is the same as for *sort*. The most interesting case is again the one for products.

The natural isomorphism $curryl : (K_1 \times K_2) \times V \cong K_1 \times (K_2 \times V)$ also known as *assoc* shifts the *least significant* part of the key into the value component. The resulting list is discriminated according to o_1 , each of the resulting groups is discriminated according to o_2 , and finally the nested groups are flattened. The definition of *discr* has an additional base case for the singleton list. This may improve the performance dramatically since the key component in the argument need not be traversed. For lexicographic string sorting this specializes to the property of MSD radix sort, which only traverses the *minimum distinguishing prefixes* of the strings, which may be substantially fewer characters than their total number. Finally, characters are sorted using bucket sort—this time we simply return the list of *non-empty* buckets.

$$\begin{aligned} & \text{bucketDiscr} :: (\text{Bounded } i, \text{Ix } i) \Rightarrow (i, i) \rightarrow [i \times v] \rightarrow [[v]] \\ & \text{bucketDiscr } \text{bs rel} \\ & = [xs \mid xs \leftarrow \text{elems } (\text{accumArray } (\lambda ws w \rightarrow ws \# [w]) [] \text{bs rel}), \neg (\text{null } xs)] \end{aligned}$$

As for *sort*, any other base type discriminator could be plugged in.

5.1 Correctness

If we concatenate the groups returned by a generic discriminator, we obtain generic sorting.

$$\text{concat} \cdot \text{discr } o = \text{sort } o \tag{5.1}$$

The proof is straightforward for units and sums, the interesting case is again the one for products. For products discrimination works from left to right, whereas sorting proceeds right to left. This means we have to be able to swap operations:

$$\begin{aligned} & \text{concat} \cdot \text{discr } (O\text{Prod } o_1 \ o_2) \\ & = \{ \text{definition of } \text{discr} \} \\ & \text{concat} \cdot \text{concat} \cdot \text{List } (\text{discr } o_2) \cdot \text{discr } o_1 \cdot \text{List } \text{curryl} \\ & = \{ \text{monad law} \} \\ & \text{concat} \cdot \text{List } \text{concat} \cdot \text{List } (\text{discr } o_2) \cdot \text{discr } o_1 \cdot \text{List } \text{curryl} \\ & = \{ \text{ex hypothesi} \} \\ & \text{concat} \cdot \text{List } (\text{sort } o_2) \cdot \text{discr } o_1 \cdot \text{List } \text{curryl} \\ & = \{ \text{proof obligation: see below} \} \\ & \text{concat} \cdot \text{discr } o_1 \cdot \text{sort } o_2 \cdot \text{List } \text{curryr} \\ & = \{ \text{ex hypothesi} \} \\ & \text{sort } o_1 \cdot \text{sort } o_2 \cdot \text{List } \text{curryr} \\ & = \{ \text{definition of } \text{sort} \} \\ & \text{sort } (O\text{Prod } o_1 \ o_2) . \end{aligned}$$

The property used in the central step, being able to swap $\text{sort } o_2$ and $\text{discr } o_1$, is actually not specific to *sort*. Our generic discriminators commute with *every strong* natural transformation. Let $\text{discr } o : [K \times V] \rightarrow [[V]]$ and $\pi : [A \times V] \rightarrow [V]$. If π is strongly natural, $\text{filter } p \cdot \pi = \pi \cdot \text{filter } (id \times p)$, then

$$\text{List } \pi \cdot \text{discr } o = \text{discr } o \cdot \pi \cdot \text{List } \text{swap} , \tag{5.2}$$

$$\begin{aligned}
 & \text{concat} \cdot \text{List} (\text{discr } o_2) \cdot \text{discr } o_1 \cdot \pi \cdot \text{List} ((\text{id} \times \text{curry}) \cdot \text{swap}) \\
 = & \{ \text{assumption: } \pi \text{ is (strongly) natural} \} \\
 & \text{concat} \cdot \text{List} (\text{discr } o_2) \cdot \text{discr } o_1 \cdot \text{List } \text{curry} \cdot \pi \cdot \text{List } \text{swap} \\
 = & \{ \text{definition of } \text{discr} \} \\
 & \text{discr} (O\text{Prod } o_1 \ o_2) \cdot \pi \cdot \text{List } \text{swap}
 \end{aligned}$$

Of course, $\text{discr } o$ itself is also strongly natural:

$$\text{List} (\text{filter } p) \cdot \text{discr } o = \text{discr } o \cdot \text{filter} (\text{id} \times p) , \quad (5.3)$$

for all $p : A \rightarrow \text{Maybe } B$. The proof is similar to the one for *sort*.

6 Generic Distributive Searching

Let us now turn to distributive searching using tries. In this paper we concentrate on bulk operations such as *trie* and *flatten*. One-at-a-time operations such as *lookup* and *insert* have been described elsewhere [5]. It turns out that *trie* is very similar to *discr*—we essentially replace $\#$ and *concat* by trie constructors. This move retains more of the original information, which is vital for supporting subsequent *efficient random* access to the values associated with a key. By storing the keys together with the values in the trie, we can even recreate all of the original keys. (If *OMap*'s key transformations are injective this can even be done without explicitly storing the keys.)

For every order representation there is a corresponding trie constructor. Additionally, we have an empty trie, which is important for efficiency reasons [5].

```

data Trie :: * -> * -> * where
    TEmpty :: Trie k v
    TUnit   :: v -> Trie () v
    TSum    :: Trie k1 v -> Trie k2 v -> Trie (k1 + k2) v
    TProd   :: Trie k1 (Trie k2 v) -> Trie (k1, k2) v
    TMap    :: (k1 -> k2) -> (Trie k2 v -> Trie k1 v)
    TChar   :: Char.Trie v -> Trie Char v
    
```

A trie of type $\text{Trie } K \ V$ represents a finite mapping from K to V , sometimes written V^K . The cases for unit, sums, and products are based on the law of exponentials: $V^1 \cong V$, $V^{K_1 + K_2} \cong V^{K_1} \times V^{K_2}$, and $V^{K_1 \times K_2} \cong (V^{K_2})^{K_1}$. The second but last case is interesting: the counterpart of *OMap* is *TMap*, which retains the key transformation. This is necessary when searching for a key that is subject to an *OMap*-order. Finally, we assume the existence of a suitable library, *Char*, implementing finite maps with character keys; for instance, character-indexed arrays, simple lists, binary trees (the basis of ternary tries). Indeed, depending on the actual data encountered, multiple data structures may even be mixed.

A trie for a given key type is a functor.

```

instance Functor (Trie k) where
    fmap f (TEmpty)   = TEmpty
    fmap f (TUnit v)  = TUnit (f v)
    
```

$$\begin{aligned}
fmap f (TSum t_1 t_2) &= TSum (fmap f t_1) (fmap f t_2) \\
fmap f (TProd t) &= TProd (fmap (fmap f) t) \\
fmap f (TMap g t) &= TMap g (fmap f t) \\
fmap f (TChar t) &= TChar (fmap f t)
\end{aligned}$$

The operation *flatten* lists the values stored in a trie.

$$\begin{aligned}
flatten &:: \text{Trie } k \ v \rightarrow [v] \\
flatten (TEmpty) &= [] \\
flatten (TUnit v) &= [v] \\
flatten (TSum t_1 t_2) &= flatten t_1 ++ flatten t_2 \\
flatten (TProd t) &= concatMap flatten (flatten t) \\
flatten (TMap g t) &= flatten t \\
flatten (TChar t) &= Char.flatten t
\end{aligned}$$

It is natural in V , that is, $flatten : \text{Trie } K \rightarrow \text{List}$.

The operation *trie* turns a finite relation, represented by an association list of type $[K \times V]$, into a finite list-valued map, represented by a trie of type $\text{Trie } K [V]$.

$$\begin{aligned}
trie &:: \text{Order } k \rightarrow [k \times v] \rightarrow \text{Trie } k [v] \\
trie \ o & \quad [] = TEmpty \\
trie \ OUnit \ rel &= TUnit (map val rel) \\
trie \ (OSum \ o_1 \ o_2) \ rel &= TSum (trie \ o_1 (filter froml rel)) (trie \ o_2 (filter fromr rel)) \\
trie \ (OProd \ o_1 \ o_2) \ rel &= TProd (fmap (trie \ o_2) (trie \ o_1 (map curryl rel))) \\
trie \ (OMap \ g \ o) \ rel &= TMap g (trie \ o (map (g \times id) rel)) \\
trie \ (OChar) \ rel &= TChar (Char.trie rel)
\end{aligned}$$

As we have noted before, *trie* arises out of *discr* by replacing $++$, *concat* etc by the appropriate trie constructors.

Indeed, if we ‘undo’ the transformation using *flatten*, we obtain the generic discriminator.

$$discr \ o = flatten \cdot trie \ o \tag{6.1}$$

The straightforward proof can be found in Appendix B.

7 Related Work

Drawing on the algorithmic techniques termed *multiset discrimination* developed by Paige and others [9], Henglein [2, 3] has shown how to make MSD distributive sorting generic by introducing generic discriminators, which have linear-time performance over a rich class of orders.

Building on the work of Connelly and Morris [10], Hinze [5] pioneered the type-indexed tries we produce. Here they are extended to support orders defined with *OMap*. The generic LSD distributive sorting and trie building functions developed here are also new. In particular, *trie* constructs a trie in bulk without incurring the substantial update costs of one-by-one insertion.

Wadler [11] derives as a “free theorem” that any function parametric in binary comparison \leq commutes with *map f* if the function *f* is an order embedding, $f x \leq' f y \iff x \leq y$; this includes all comparison-based sorting algorithms. As shown by Day et al. [12], Knuth’s *0-1 principle* for sorting networks [13] can also be seen as a free theorem for sorting networks formulated as comparator-parametrized functions. These properties correspond to naturality, respectively parametricity properties on the keys; they are *different* from our naturality property 4.3 since the latter applies to the value components and leaves the key components invariant.

Our strong naturality property 4.4, coupled with preserving singletons and correct sorting of two-element lists, corresponds to Henglein’s *consistent permutativity*, which characterizes stable sorting functions [14]. It is, however, a more general and a more elegant formulation supporting equational reasoning. In particular, it highlights the semantic benefits of adopting a formulation for sorting based on key-value pairs rather than keys alone.

Gibbons [15] shows how an LSD radix sort for lists can be derived from a stable MSD radix sort that first builds an explicit trie and then flattens it into the result list. Since MSD radix sort, even with explicit tries, is sometimes preferable to LSD radix sort (to avoid sparse bucket table traversal [16] and for large data sets [17]), the derivation makes sense in both directions. Our development can be seen as a generalization of Gibbons’ work: it works for arbitrary denotable orders over any type; we decompose MSD sorting into discrimination followed by concatenation, without the need for a trie (though it can be achieved by way of *trie*); and our commutativity property 5.2 holds for *any* strong natural transformation, not just for sorting functions.

8 Conclusion

Comparison-based sorting algorithms and search trees are easily made generic, that is, applicable to user-defined orders, by abstraction over the comparison function. This has arguably contributed to their popularity even though distributive (radix/trie) and hashing techniques often have superior performance for special types, such as machine integers and character strings.

We have shown how to construct generic comparison, sorting, discrimination and trie building operations by induction over a class of orders including standard orders on primitive types; lexicographic orders on sums, products and lists; and orders defined as the inverse image of a given order under an arbitrary function.

We have identified strong naturality—commutativity with filtering—as a powerful property of stable sorting functions, and shown discrimination to commute with any strongly natural transformation, including, but not limited to, stable sorting functions.

The trie building operation yields a data structure that is not only asymptotically as efficient as discrimination but also supports efficient key-based random access, without incurring a one-at-a-time insertion overhead during construction.

Future work consists of extending equational reasoning and calculational correctness proofs to coinductive order representations; investigating *data-dependent* variations of our generic functions and staged execution for our *data-independent* generic functions by compile-time specialization (partial evaluation) and exploiting parallelism at word, multicore, and manycore/GPU levels; and eventually providing architecture-independent frameworks encapsulating distributive sorting and searching methods as semantically (obeying representation independence) and computationally (exhibiting superior performance) well-behaved alternatives to comparison-based methods.

Acknowledgements

We would like to thank the anonymous referees of APLAS 2013 and Nicolas Wu for suggesting various presentational improvements. We owe a particular debt of gratitude to Richard Bird for carefully reading a draft version of this paper, pointing out typographical errors, glitches of language, and for suggesting pronounceable identifiers. This work has been funded by EPSRC grant number EP/J010995/1 and Danish Research Council grant 10-092299 for HIPERFIT.

References

1. Strachey, C.: Fundamental concepts in programming languages. *Higher-order and symbolic computation* **13**(1) (2000) 11–49
2. Henglein, F.: Generic discrimination: Sorting and partitioning unshared data in linear time. In Hook, J., Thiemann, P., eds.: *Proc. 13th ACM SIGPLAN Int'l Conf. on Functional Programming (ICFP)*, ACM (September 2008) 91–102
3. Henglein, F.: Generic top-down discrimination for sorting and partitioning in linear time. *Journal of Functional Programming (JFP)* **22**(3) (July 2012) 300–374
4. Bentley, J.L., Sedgwick, R.: Fast algorithms for sorting and searching strings. In: *SODA '97: Proceedings of the eighth annual ACM-SIAM symposium on Discrete algorithms*, Philadelphia, PA, USA, Society for Industrial and Applied Mathematics (1997) 360–369
5. Hinze, R.: Generalizing generalized tries. *Journal of Functional Programming* **10**(4) (2000) 327–351
6. Hoare, C.A.: Quicksort. *The Computer Journal* **5**(1) (1962) 10–16
7. Bayer, R.: Symmetric binary B-trees: Data structure and maintenance algorithms. *Acta informatica* **1**(4) (1972) 290–306
8. Guibas, L.J., Sedgwick, R.: A dichromatic framework for balanced trees. In: *Proc. 19th Annual Symposium on Foundations of Computer Science (FOCS)*, IEEE (1978) 8–21
9. Cai, J., Paige, R.: Using multiset discrimination to solve language processing problems without hashing. *Theoretical Computer Science (TCS)* **145**(1-2) (July 1995) 189–228
10. Connelly, R.H., Morris, F.L.: A generalization of the trie data structure. *Mathematical Structures in Computer Science* **5**(3) (September 1995) 381–418
11. Wadler, P.: Theorems for free! In: *Proc. Functional Programming Languages and Computer Architecture (FPCA)*, London, England, ACM Press (September 1989) 347–359

12. Day, N.A., Launchbury, J., Lewis, J.: Logical abstractions in haskell. In: Proc. Haskell Workshop. Number UU-CS-1999-28 in Technical Report, Utrecht, The Netherlands, Utrecht University (1999)
13. Knuth, D.: The Art of Computer Programming: Sorting and Searching. 2nd edn. Volume 3. Addison Wesley (1998)
14. Henglein, F.: What is a sorting function? J. Logic and Algebraic Programming (JLAP) **78**(5) (May-June 2009) 381–401 Invited submission to special issue on 19th Nordic Workshop on Programming Theory (NWPT).
15. Gibbons, J.: A pointless derivation of radix sort. Journal of Functional Programming **9**(3) (1999) 339–346
16. Paige, R., Tarjan, R.E.: Three partition refinement algorithms. SIAM Journal of Computing **16**(6) (December 1987) 973–989
17. Sinha, R., Zobel, J.: Efficient trie-based sorting of large sets of strings. In: Proc. 26th Australasian Computer Science Conference (ACSC). (2003) 11–18

A Proof of Property (4.6)

The cases for unit and sums are straightforward. We only consider the product case, which is actually instructive.

Case $o = OProd\ o_1\ o_2$: again, we make use of naturality to be able to apply the induction assumption.

$$\begin{aligned}
 & \text{sort } (OProd\ o_1\ o_2) [((a_1, a_2), i), ((b_1, b_2), j)] = [i, j] \\
 \iff & \{ \text{definition of } \text{sort} \} \\
 & \text{sort } o_1 (\text{sort } o_2 (\text{List } \text{curryr} [((a_1, a_2), i), ((b_1, b_2), j)])) = [i, j] \\
 \iff & \{ \text{definition of } \text{curryr} \} \\
 & \text{sort } o_1 (\text{sort } o_2 [(a_2, (a_1, i)), (b_2, (b_1, j))]) = [i, j] \\
 \iff & \{ \text{let } re\ i = (a_1, i) \text{ and } re\ j = (b_1, j) \} \\
 & \text{sort } o_1 (\text{sort } o_2 (\text{List } (id \times re) [(a_2, i), (b_2, j)])) = [i, j] \\
 \iff & \{ \text{sort is natural (4.3)} \} \\
 & \text{sort } o_1 (\text{List } re (\text{sort } o_2 [(a_2, i), (b_2, j)])) = [i, j]
 \end{aligned}$$

The strict version of leq suggests to conduct a case analysis on $leq\ o_2\ a_2\ b_2$.

Case $leq\ o_2\ a_2\ b_2$:

$$\begin{aligned}
 \iff & \{ \text{ex hypothesi} \} \\
 & \text{sort } o_1 (\text{List } re [i, j]) = [i, j] \\
 \iff & \{ \text{definition of } re \} \\
 & \text{sort } o_1 [(a_1, i), (b_1, j)] = [i, j] \\
 \iff & \{ \text{ex hypothesi} \} \\
 & leq\ o_1\ a_1\ b_1 \\
 \iff & \{ \text{definition of } leq \} \\
 & leq (OProd\ o_1\ o_2) (a_1, a_2) (b_1, b_2)
 \end{aligned}$$

Case $\neg (leq\ o_2\ a_2\ b_2)$:

$$\begin{aligned}
 \iff & \{ \text{ex hypothesi} \} \\
 & \text{sort } o_1 (\text{List } re [j, i]) = [i, j] \\
 \iff & \{ \text{definition of } re \} \\
 & \text{sort } o_1 [(b_1, j), (a_1, i)] = [i, j] \\
 \iff & \{ \text{ex hypothesi} \} \\
 & \neg (leq\ o_1\ b_1\ a_1) \\
 \iff & \{ \text{definition of } leq \} \\
 & leq (OProd\ o_1\ o_2) (a_1, a_2) (b_1, b_2)
 \end{aligned}$$

B Proof of $\mathit{discr} \ o = \mathit{flatten} \cdot \mathit{trie} \ o$ (6.1)**Case $o = OUnit$:**

$$\begin{aligned}
& \mathit{flatten} \cdot \mathit{trie} \ OUnit \\
= & \{ \text{definition of } \mathit{trie} \} \\
& \mathit{flatten} \cdot TUnit \cdot \mathit{map} \ \mathit{val} \\
= & \{ \text{definition of } \mathit{flatten} \} \\
& \mathit{return} \cdot \mathit{map} \ \mathit{val} \\
= & \{ \text{definition of } \mathit{discr} \} \\
& \mathit{discr} \ OUnit
\end{aligned}$$

Case $o = OSum \ o_1 \ o_2$:

$$\begin{aligned}
& \mathit{flatten} \cdot \mathit{trie} \ (OSum \ o_1 \ o_2) \\
= & \{ \text{definition of } \mathit{trie} \} \\
& \mathit{flatten} \cdot TSum \cdot (\mathit{trie} \ o_1 \cdot \mathit{filter} \ (\mathit{inl}^\circ \times \mathit{id}) \ \Delta \ \dots) \\
= & \{ \text{definition of } \mathit{flatten} \} \\
& (\mathit{flatten} \cdot \mathit{outl} \ \# \ \dots) \cdot (\mathit{trie} \ o_1 \cdot \mathit{filter} \ (\mathit{inl}^\circ \times \mathit{id}) \ \Delta \ \dots) \\
= & \{ \text{fusion and computation} \} \\
& \mathit{flatten} \cdot \mathit{trie} \ o_1 \cdot \mathit{filter} \ (\mathit{inl}^\circ \times \mathit{id}) \ \# \ \dots \\
= & \{ \text{ex hypothesi} \} \\
& \mathit{discr} \ o_1 \cdot \mathit{filter} \ (\mathit{inl}^\circ \times \mathit{id}) \ \# \ \dots \\
= & \{ \text{definition of } \mathit{discr} \} \\
& \mathit{discr} \ (OSum \ o_1 \ o_2)
\end{aligned}$$

Case $o = OProd \ o_1 \ o_2$: here we make essential use of the fact that $\mathit{Trie} \ K$ is a functor and that $\mathit{flatten}$ is natural in V .

$$\begin{aligned}
& \mathit{flatten} \cdot \mathit{trie} \ (OProd \ o_1 \ o_2) \\
= & \{ \text{definition of } \mathit{trie} \} \\
& \mathit{flatten} \cdot TProd \cdot \mathit{Trie} \ K_1 \ (\mathit{trie} \ o_2) \cdot \mathit{trie} \ o_1 \cdot \mathit{List} \ \mathit{curryl} \\
= & \{ \text{definition of } \mathit{flatten} \} \\
& \mathit{concat} \cdot \mathit{List} \ \mathit{flatten} \cdot \mathit{flatten} \cdot \mathit{Trie} \ K_1 \ (\mathit{trie} \ o_2) \cdot \mathit{trie} \ o_1 \cdot \mathit{List} \ \mathit{curryl} \\
= & \{ \mathit{flatten} \text{ is natural: } \mathit{List} \ f \cdot \mathit{flatten} = \mathit{flatten} \cdot \mathit{Trie} \ K \ f \} \\
& \mathit{concat} \cdot \mathit{List} \ \mathit{flatten} \cdot \mathit{List} \ (\mathit{trie} \ o_2) \cdot \mathit{flatten} \cdot \mathit{trie} \ o_1 \cdot \mathit{List} \ \mathit{curryl} \\
= & \{ \text{ex hypothesi, twice} \} \\
& \mathit{concat} \cdot \mathit{List} \ (\mathit{discr} \ o_2) \cdot \mathit{discr} \ o_1 \cdot \mathit{List} \ \mathit{curryl} \\
= & \{ \text{definition of } \mathit{discr} \} \\
& \mathit{discr} \ (OProd \ o_1 \ o_2) \ .
\end{aligned}$$